

** Trump 604
** Proposal title: Does inequality activate the system justification motivation.

/*
HYPOTHESES

Overview: The study has a "high" vs. "low" inequality experimental treatment
and then three outcome scales.  The basic idea is that exposure to the high
inequalitytreatment will increase system justification.

Test-Hyp1: The "high" inequality treatment will increase response to the
system justification scale.

Test-Hyp2: The "high" inequalty treatment will increase response to questions
about trust in major institutions.

Test-Hyp3: The "high" inequality treatment will increase response to questions
about economic system justification.

Stated-Hyp: "Following Jost, Banaji, and Nosek, we expect the impact of the
'high' inequality treatment to be particularly strong on the system justification
motivation of low-income individuals, and that the effect among the relatively
disadvantaged will be strong enough to result in a direct effect of the treatment."

Test-Hyp4: The treatment effect on system justification scale will be larger
for lower income respondents.

Test-Hyp5: The treatment effect on confidence in major institutions will be larger
for lower income respondents.

Test-Hyp6: The treatment effect on economic system justification will be larger
for lower income respondents.

*/

version 15

/*
** import and convert data to stata format
import spss using "TESS3 174 - Trump_Client.sav", clear
save trump605, replace
*/

********************************************************************************

* CONSTRUCT INDICATORS OF EXPERIMENTAL MANIPULATIONS
clear all
use trump605, clear

* simple two-group experiment, although divided into
* six groups to determine which outcome measures are asked.
recode TESS174 (1/3=1) (4/6=0) (*=.), gen(ineq_rising)

** OUTCOME MEASURES

** scale a -- system justification scale
	* (coded so higher values -> system justification)

forvalues i = 1(1)8 {
	gen scalea_`i' = QA_`i' if QA_`i' != -1
}

gen scalea = (10-scalea_1) + (10-scalea_2) + scalea_3 + (10-scalea_4) + ///
		(10-scalea_5) + (10-scalea_6) + scalea_7 + (10-scalea_8)
	

forvalues i = 1(1)6 {
	clonevar scaleb_`i' = QB_`i' if QB_`i' != -1
}

** scale b -- confidence in social institutions
	* (coded so higher values -> more confidence)

gen scaleb = (4-scaleb_1) + (4-scaleb_2) + (4-scaleb_3) + ///
		(4-scaleb_4) + (4-scaleb_5) + (4-scaleb_6)
		
		
** scale c -- economic system justification scale 
	* (coded so higher values -> system justification)

forvalues i = 1(1)15 {
	gen scalec_`i' = QC_`i' if QC_`i' != -1
}

gen scalec = (10-scalec_1) + scalec_2 + (10-scalec_3) + scalec_4 + ///
	(10-scalec_5) + scalec_6 + (10-scalec_7) + scalec_8 + ///
	scalec_9 + (10-scalec_10) + (10-scalec_11) + scalec_12 + ///
	(10-scalec_13) + scalec_14 + (10-scalec_15)

	
** MODERATOR

* Note: the proposal indicates moderation by low-income but does provide
* a cutoff.  Specifying income as a continuous moderator.

* income
	* midpoints: top-coded highest category to 25% over top-code 
	
recode PPINCIMP (1=2.5) (2=6.25) (3=8.75) (4=11.25) (5=13.5) ///
	(6=17.5) (7=22.5) (8=27.5) (9=32.5) (10=37.5) (11=45) ///
	(12=55) (13=67.5) (14=80) (15=92.5) (16=112.5) ///
	(17=137.5) (18=162.5) (19=218.75), gen(hhincome)

	
	
********************************************************************************

* CONSTRUCT INDICATORS OF EXPERIMENTAL MANIPULATIONS


/*
Test-Hyp1: The "high" inequality treatment will increase response to the
system justification scale.
*/	
regress scalea 1.ineq_rising
// NOT SUPPORTED - p = .026 (Bonferroni corrected threshold)
tess 1.ineq_rising +, bonf(3) init(Trump605)

/*
Test-Hyp2: The "high" inequalty treatment will increase response to questions
about trust in major institutions.
*/
regress scaleb 1.ineq_rising
// NOT SUPPORTED - p = .194 (but wrong direction)
tess 1.ineq_rising +, bonf(3)

/*
Test-Hyp3: The "high" inequality treatment will increase response to questions
about economic system justification.
*/	
regress scalec 1.ineq_rising
// NOT SUPPORTED - p = .074 (correct direction)
tess 1.ineq_rising +, bonf(3)

** moderator hypotheses

/*
Test-Hyp4: The treatment effect on system justification scale will be larger
for lower income respondents.
*/
regress scalea i.ineq_rising##c.hhincome
// NOT SUPPORTED - p = .177 
tess 1.ineq_rising#c.hhincome +, bonf(3)

/*
Test-Hyp5: The treatment effect on confidence in major institutions will be larger
for lower income respondents.
*/
regress scaleb i.ineq_rising##c.hhincome
// NOT SUPPORTED - p = .995 (wrong direction)
tess 1.ineq_rising#c.hhincome +, bonf(3)

/*
Test-Hyp6: The treatment effect on economic system justification will be larger
for lower income respondents.
*/	
regress scalec i.ineq_rising##c.hhincome
// NOT SUPPORTED - p = .071 (correct direction)
tess 1.ineq_rising#c.hhincome +, bonf(3)

